
*** Read and prepare original data

foreach u in under75 above75 fine { //

	if "`u'"=="under75" import excel "${regionalstatistik_in}12411-02-03-4.xlsx", cellrange(A6) clear
	if "`u'"=="above75" import excel "${regionalstatistik_in}12411-04-02-4.xlsx", cellrange(A6) clear
	if "`u'"=="fine" import excel "${regionalstatistik_in}12411-04-02-4_fine.xlsx", cellrange(A6) clear

	* varnames 

	rename A kreis 
	rename B kreisname
	rename C altersgruppe 
	rename D pop_tot
	rename E pop_m
	rename F pop_f

	* replace missings

	replace pop_tot = "" if pop_tot == "-"
	replace pop_m = "" if pop_m == "-"
	replace pop_f = "" if pop_f == "-"

	* Kreise
	drop if altersgruppe == ""
	
	replace kreis = strtrim(kreis)
	replace kreisname = strtrim(kreisname)

	replace kreisname = kreisname[_n-1] if kreisname == "" 
	replace kreis = kreis[_n-1] if kreis == ""


	* reshape prep
	gen alter = ""

	if "`u'"=="fine" {
	
		replace alter = "3_4" if altersgruppe == "3 bis unter 4 Jahre"
		replace alter = "4_5" if altersgruppe == "4 bis unter 5 Jahre"
		replace alter = "5_6" if altersgruppe == "5 bis unter 6 Jahre"
	
	} 
	else {
	
		replace alter = "0_3" if altersgruppe == "unter 3 Jahre"
		replace alter = "6_10" if altersgruppe == "6 bis unter 10 Jahre"
		replace alter = "10_15" if altersgruppe == "10 bis unter 15 Jahre"
		replace alter = "15_18" if altersgruppe == "15 bis unter 18 Jahre"
		replace alter = "18_20" if altersgruppe == "18 bis unter 20 Jahre"
		replace alter = "20_25" if altersgruppe == "20 bis unter 25 Jahre"
		replace alter = "25_30" if altersgruppe == "25 bis unter 30 Jahre"
		replace alter = "30_35" if altersgruppe == "30 bis unter 35 Jahre"
		replace alter = "35_40" if altersgruppe == "35 bis unter 40 Jahre"
		replace alter = "40_45" if altersgruppe == "40 bis unter 45 Jahre"
		replace alter = "45_50" if altersgruppe == "45 bis unter 50 Jahre"
		replace alter = "50_55" if altersgruppe == "50 bis unter 55 Jahre"
		replace alter = "55_60" if altersgruppe == "55 bis unter 60 Jahre"
		replace alter = "60_65" if altersgruppe == "60 bis unter 65 Jahre"
		replace alter = "65_75" if altersgruppe == "65 bis unter 75 Jahre"
		replace alter = "75_80" if altersgruppe == "75 bis unter 80 Jahre"
		replace alter = "80_85" if altersgruppe == "80 bis unter 85 Jahre"
		replace alter = "85_90" if altersgruppe == "85 bis unter 90 Jahre"
		replace alter = "90plus" if altersgruppe == "90 Jahre und mehr"
	
	}

	drop altersgruppe
	
	keep if alter != ""
	
	destring pop_*, replace 
	
	* Kreis ID
	gen length = length(kreis)

	drop if length != 5 & kreis != "02" & kreis != "11" 

	replace kreis = "02000" if kreis=="02"
	replace kreis = "11000" if kreis=="11"	

	drop length
	
	destring kreis, replace
	rename kreis kreis_id
	
	qui compress
	save "${regionalstatistik_out}population_`u'.dta", replace	
	
}


clear all
append using "${regionalstatistik_out}population_under75.dta"
append using "${regionalstatistik_out}population_above75.dta"
append using "${regionalstatistik_out}population_fine.dta"



* age groups
	gen age 	= 0 if inlist(alter,"0_3","3_4","4_5")
	replace age = 1 if inlist(alter,"5_6","6_10","10_15")
	replace age = 2 if inlist(alter,"15_18","18_20","20_25","25_30","30_35") 
	replace age = 3 if inlist(alter,"35_40","40_45","45_50","50_55","55_60")
	replace age = 4 if inlist(alter,"60_65","65_75","75_80","80_85","85_90","90plus")

	collapse (sum) pop_? if age!=., by(kreis_id age)
* reshape 
	reshape long pop , i(kreis_id age) j(gender) string

	gen female = (gender=="_f")
	drop gender
	
* save
	qui compress
	rename kreis_id idlandkreis
	save "${regionalstatistik_out}kreis_pop_age_gender_Destatis.dta", replace 
	